



Einfach, schnell & aussagefähig – ODS Graphics in SAS9.2

sas forum
SWITZERLAND 2009 baden

Hans-Rainer Pauli
Solution Specialist Technology
SAS Institute


Agenda

- Was ist ODS Graphics?
- ODS Statistical Graphics
- Neue ODS Prozeduren
- Graph Template Language (GTL)
- Interaktive Werkzeuge
- Q & A

Was ist ODS Graphics?

- von ansprechender, hoher Qualität
- hat sprachbasierende Werkzeuge für Programmierer
- hat interaktive Werkzeuge für jedermann
- EINFACH !

Ein „einfaches“ Beispiel



Was ist ODS Graphics?

- ODS Graphics besteht aus folgenden Komponenten:
 - ODS Statistical Graphics
 - Neue ODS Prozeduren
 - Graph Template Language (GTL)

 - ODS Graphics Editor
 - ODS Graphics Designer (pre-production)

ODS Graphics

ODS Statistical Graphics

- Neue Funktionalität für die Erstellung statistischer Graphiken
- Verfügbar in über 50 Prozeduren (Base SAS, SAS/STAT, SAS/ETS, SAS/QC, ...)
 - Benötigt SAS/GRAPH
- Kundenreaktionen
 - “compelling reason for us to move to SAS9.2”
 - “now, we don’t have to do our graphics in S-Plus anymore”

ODS Graphics

ODS Statistical Graphics

*The REG Procedure
 Model: MODEL1
 Dependent Variable: Weight*

Parameter Estimates					
Variable	DF	Parameter Estimate	Standard Error	t Value	Pr > t
Intercept	1	-143.02692	32.27459	-4.43	0.0004
Height	1	3.89903	0.51609	7.55	<.0001

ODS in SAS Version 8

ods graphics on;







```
proc reg data=sashelp.class;
    model height=weight;
run;quit;
```

ods graphics off;

ODS Graphics in SAS9.2

ODS Graphics




ODS Statistical Graphics



ODS Graphics

ODS Statistical Graphics

- Support für Styles
 - Ein ODS Style kontrolliert, wie Graphiken generell aussehen
 - Styles sind optimiert für analytische Darstellungen
 - Erzeugen ästhetisch ansprechende Darstellungen “out of the box”



ODS Graphics

Neue ODS Prozeduren

- Vier neue ODS Prozeduren in SAS/GRAPH
 - PROC SGPLOT
 - PROC SGPANEL
 - PROC SGSCATTER
 - PROC SGRENDER
- Sprache ähnlich wie andere SAS/GRAPH Prozeduren
- Für Analysten und SAS Programmierer

ODS Graphics


"Single-Cell" Graphiken: PROC SGPLOT

```

title "Distribution of Fuel Consumption";
proc sgplot data=sashelp.cars;
  histogram mpg_city;
  density mpg_city;
  density mpg_city / type=kernel;
  keylegend / location = inside
            position=TopRight
            across=1;


  yaxis grid;
run;

```



ODS Graphics


"Single-Cell" Graphiken: PROC SGPLOT



ODS Graphics


"Single-Cell" Graphiken: PROC SGPLOT

```
proc sgplot data=response;
  axis integer values=(0 to 15 by 3)
    label="Weeks in Treatment";
  yaxis label="Percent Change";
  band y=ymax lower=12.1 upper=15 /
    transparency=.8 fillattrs=graphdata1;
  scatter x=week y=pct / group=trt
    yerrorlower=lo yerrorupper=hi
    markerattrs=(symbol=circlefilled)
    name="scat";
  series x=week y=pct / group=trt
    lineattrs=(pattern=solid);
  series x=week y=end / group=trt
    lineattrs=(pattern=shortdash)
    markers markerattrs=(symbol=circle);
  refline 0;
  refline 13.5 / axis=x lineattrs=(thickness=0px)
    label="|-- Washout --|" labelloc=outside
    labelpos=min;
  keylegend "scat" / title="" noborder;
run;
```



ODS Graphics


"Single-Cell" Graphiken: PROC SGPLOT



ODS Graphics

"Single-Cell" Graphiken: PROC SGPLOT

```
proc sgplot data=cancer;
  format mcases mdeaths fcases fdeaths positive.;
  hbar cause / response=mcases
    fillattrs=graphdata1 transparency=.65
    legendlabel="New Cases (Male)"
    name="mcases" ;
  hbar cause / response=mdeaths barwidth=.5
    fillattrs=graphdata1 transparency=.25
    legendlabel="Deaths (Male)"
    name="mdeaths" ;
  hbar cause / response=fcases
    fillattrs=graphdata2 transparency=.65
    legendlabel="New Cases (Female)"
    name="fcases";
  hbar cause / response=fdeaths barwidth=.5
    fillattrs=graphdata2 transparency=.25
    legendlabel="Deaths (Female)"
    name="fdeaths";
  keylegend "mcases" "fcases" "mdeaths" "fdeaths"/
  across=2;
  yaxis label=" " discreteorder=data;
  xaxis label=" " grid;
run;
```



ODS Graphics

“Multi-Cell” Graphiken

- Panels mit PROC SGPANEL
 - Panel von “Single-Cell” Plots mit einer oder mehr Klassifizierungsvariablen
 - Angeordnet als LATTICE, PANEL, COLUMNLATTICE, oder ROWLATTICE
 - Gemeinsame Achsen und Legenden
- Scatter Plots mit PROC SGSCATTER
 - Unabhängige Scatter Plots
 - Scatter Plot Matrizen
 - Gemeinsame oder unabhängige Achsen
 - Gemeinsame Legende


ODS Graphics

"Multi-Cell" Graphiken: PROC SGPANEL

```

title "Barley Yield at Six Sites";
proc sgpanel data=barley;
  panelby site/novarname;
  dot variety/response=yield
        group=year;
  discretelegend;
run;

```




ODS Graphics

"Multi-Cell" Graphiken: PROC SGPANEL

```

title 'Yearly Sales Figures by Country and Product';
proc sgpanel data=sashelp.prdsale;
  where product in ('BED','CHAIR','DESK');
  panelby country product /
  layout=lattice;
  hbox actual /
    category=year ;
  rowaxis display=(nolabel);
run;

```



ODS Graphics


"Multi-Cell" Graphiken: PROC SGSCATTER

Unabhängige Scatter Plots

```

title "Car Performance Data";
proc sgscatter data=cars;
plot (mpg_city weight) *
      (horsepower displacement)/
      group=type loess grid;
run;

```




ODS Graphics

"Multi-Cell" Graphiken: PROC SGSCATTER

Scatter Plot Matrix

```
proc sgscatter data=heart;
  matrix systolic diastolic
        cholesterol /
  ellipse=(alpha=0.05 type=predicted)
  diagonal=(histogram normal);
run;
```



ODS Graphics

The Graph Template Language (GTL)

- Ein Template beschreibt das Aussehen individueller Graphiken, welches mittels der Graph Template Language (GTL) geschrieben wird
- SAS stellt für alle Graphiken, welche automatisch von bestimmten Prozeduren erstellt werden, ein Template zur Verfügung
- Diese Templates können modifiziert werden, um eine gewünschte Änderung permanent verfügbar zu machen
- SAS Programmierer können mit der GTL individuelle Templates für komplexe Darstellungen erstellen

ODS Graphics

Templates


- Die Graph Template Language
 - ist eine Erweiterung der TEMPLATE Prozedur
 - verbindet Layouts und Plots auf flexible Art und Weise
 - ermöglicht statistische Berechnungen und Darstellungen
 - verwendet ODS Styles
 - ist ein mächtiges Werkzeug für die Erstellung komplexer Graphiken
 - wird auch für “build-in” Graphiken und ODS Prozeduren verwendet

ODS Graphics

Graph Template Language

```

begingraph;
  entrytitle "Class Profile";
  layout lattice / rowweight=(0.5 0.2 0.3)
                 rowgutter=5
                 columndatarange=union;
  ( layout overlay / xaxisopts=(display=(line));
    histogram weight;
    densityplot weight / normal();
  endlayout;
  ( layout overlay / xaxisopts=(display=(line));
    boxplot y=weight / orient=horizontal;
  endlayout;
  ( layout overlay / xaxisopts=(label="Weight");
    fringeplot weight;
  endlayout;
endlayout;
endgraph;
    
```



ODS Graphics

Graph Template Language

- Dynamische Elemente in Templates
 - Templates können dynamische Elemente enthalten, welches eine allgemeinere Nutzung ermöglicht
- Dynamische Referenzen in PROC SGRENDER
- Macro-Variablen
- Konditionale Logik für dynamische Optionen

ODS Graphics

Graph Template Language: Dynamische Elemente

```


proc template;
  define statgraph dynamics;
    begingraph;
      mvar SYSDATE9 SCALE;
      nmvar BINS;
      dynamic VAR VARLABEL;
      entrytitle "Histogram of " VAR;
      entrytitle "with Normal Distribution";
      layout overlay / xaxisopts=(label=VARLABEL);
        histogram VAR / scale=SCALE nbins=BINS;
        densityplot VAR / normal();
      endlayout;
      entryfootnote halign=right "Created: " SYSDATE9 /
        textattrs=GraphValueText;
    endgraph;
  end;
run;

```

ODS Graphics

Graph Template Language: Dynamische Elemente


```
%let bins=7;  
%let scale=count;  
  
proc sgrender  
  data=sashelp.class  
  template=dynamics;  
  
  dynamic var="Height"  
          varlabel="Height in  
          Inches";  
  
run;
```



ODS Graphics

Graph Template Language: Dynamische Elemente

```
%let bins=5;  
%let scale=percent;  
  
proc sgrender data=sashelp.class  
  template=dynamics;  
  
  dynamic var="Weight"  
          varlabel="Weight in Pounds";  
  
run;
```



ODS Graphics

Interaktive Werkzeuge

- ODS Graphics Editor
 - Interaktiver Editor um Output von Prozeduren anzupassen

- ODS Graphics Designer (pre-production)
 - Interaktives Werkzeug um analytische Graphiken von Grund auf ohne Programmierkenntnisse zu erstellen


ODS Graphics

ODS Graphics Editor

- Titel und Fussnoten editieren
- Marker, Linien, Schriftarten, etc. anpassen
- Annotationen hinzufügen
- “Copy and Paste”
- Angepasste Darstellung abspeichern

ODS Graphics


ODS Graphics Editor



Output from
 ROBUSTREG

ODS Graphics

ODS Graphics Editor



Title Changed
 and Outlier
 Highlighted


ODS Graphics

ODS Graphics Designer (pre-production)


- Ermöglicht die interaktive Erstellung von ODS Graphiken via “drag-and-drop”
- Erzeugt GTL Programmcode, der mit PROC TEMPLATE weiterverwendet werden kann
- Graphik kann als .SGD File abgespeichert werden und in “batch” mit PROC SGDESIGN ausgeführt werden

ODS Graphics

ODS Graphics Designer (pre-production)



Ein einfaches Beispiel



Task	What Do You Use?	What Is Involved?
Creating graphs for statistical Analyses	Statistical procedures that support ODS Graphics	ODS GRAPHICS ON statement; graphs created by default or with procedure options
Creating stand-alone graphs for exploration of data or for customized Displays	SAS/GRAPH SGPLOT, SGPANEL, SGSCATTER procedures	Procedure Syntax
Changing overall appearance of graphs and tables	ODS styles	STYLE= option in ODS destination statement
Enhancing specific graphs for presentation or paper	ODS Graphics Editor	Request editable graphs, invoke Editor, then use point-and-click interface
Making persistent, programmatic changes in graphs	Default ODS graph template supplied by the SAS System	Modify default graph template with Graph Template Language and compile with TEMPLATE procedure
Creating highly customized graphs	User-written graph template	Write template with Graph Template Language, compile with TEMPLATE procedure, then apply to data with SGRENDER procedure


For More Information...

- Base focus area (with TS Samples and SAS Notes)
(<http://support.sas.com/rnd/base/ods/index.html>)
- SAS/GRAPH 9.2 Manuals
 - SG Procedures Guide, GTL User's Guide, GTL Reference, ODS Graphics Editor User's Guide, SAS/STAT User's Guide
- SAS Global Forum, 2009
 - Secrets of the SG Procedures
(<http://support.sas.com/resources/papers/proceedings09/324-2009.pdf>)
 - ODS Graphics Designer
(<http://support.sas.com/resources/papers/proceedings09/198-2009.pdf>
<http://support.sas.com/resources/papers/proceedings09/331-2009.pdf>)
 - Clinical Trial Reporting Using SAS/GRAPH® SG Procedures
(<http://support.sas.com/resources/papers/proceedings09/174-2009.pdf>)

For More Information...



- SAS Global Forum, 2008
 - Effective Graphics Made Simple using SAS/GRAPH “SG” Procedures
(<http://www2.sas.com/proceedings/forum2008/255-2008.pdf>)
 - ODS Graphics Editor
(<http://www2.sas.com/proceedings/forum2008/235-2008.pdf>)
 - Butterflies, Heat Maps, and More
(<http://www2.sas.com/proceedings/forum2008/243-2008.pdf>)
- SAS Global Forum, 2007
 - New SAS/GRAPH Procedures for Creating Statistical Graphics
(<http://www2.sas.com/proceedings/forum2007/193-2007.pdf>)
- SUGI 31, 2006
 - Creating Statistical Graphics in SAS9.2: What Every Statistical User Should Know
<http://www2.sas.com/proceedings/sugi31/192-31.pdf>

SAS Graph Evolution



SAS 8

Principal Component Scores Matrix Plot



Q & A



The logo features the text 'sas forum' in a white, lowercase, sans-serif font. The 'sas' part is bolded, and a registered trademark symbol (®) is positioned to the upper right of the 's'. Below this, 'SWITZERLAND 2009 baden' is written in a smaller, all-caps, sans-serif font. The 'baden' part is bolded. The background is a dark blue gradient with a bright, multi-pointed starburst light in the upper right quadrant and faint, abstract white lines and dots scattered throughout.

sas forum
SWITZERLAND 2009 **baden**